
Ohsaki’s Spectrum―OHSP 
 
 
 The program OHSP (Ohsaki's Spectrum) is a subroutine subprogram that calculates the response velocity 

(spectral velocity) for an arbitrary period to fit the Ohsaki spectrum defined by the control points A, B, C, 

D, and E in the table below for given the earthquake magnitude, epicentral distance, and damping factor. 
 
 

Table Coordinates of the control point 

 

 
 

OHSP（Ohsaki's Spectrum） 
 
【Purpose】 
 To compute the spectral velocity in the Ohsaki’s spectrum for a given period when the magnitude and 
epicentral distance of an earthquake and the damping factor of the spectrum are specified. 
 
 
【Usage】 
（1） How to connect 
    CALL OHSP(EM, R, H, T, SV, ICALL)  
 

T A (S v)A T B (S v)B T C (S v)C T D (S v)D T E (S v)E

6 5 1.22 0.10 15 0.14 21 0.80 21 9

7 10 0.52 0.20 12 0.40 25 1.20 25 20

8 25 0.28 0.35 4 0.60 30 1.50 30 28

6 15 0.76 0.12 9 0.25 25 0.65 25 12

7 45 0.42 0.28 10 0.45 27 0.90 27 24

8 120 0.30 0.45 10 0.70 33 1.20 33 32

6 60 0.62 0.14 8 0.32 27 0.60 27 15

7 150 0.46 0.28 11 0.50 32 0.80 32 27

8 350 0.36 0.50 18 0.70 38 1.00 38 37

 (2） Necessary subroutines and function subprograms 
VELK 

 
 
【Program List】 
 

C  * * * * * * * * * * * * * * * * * * * *                              OHSP   1 

C     SUBROUTINE FOR OHSAKI'S SPECTRUM                                  OHSP   2 

C  * * * * * * * * * * * * * * * * * * * *                              OHSP   3 

C                                                                       OHSP   4 

C                    CODED BY Y.OHSAKI                                  OHSP   5 

C                                                                       OHSP   6 

C     PUBPOSE                                                           OHSP   7 

C       TO COMPUTE, FOR A GIVEN PERIOD, THE SPECTRL VELOCITY IN         OHSP   8 

C       OHSAKI'S SPECTRUM WHEN THE MAGNITUDE AND EPICENTRAL DISTANCE    OHSP   9 

C       OF AN EARTHQUAKE AND THE DAMPING FACTOR OF THE SPECTRUM ARE     OHSP  10 

C       SPECIFIED                                                       OHSP  11 

C                                                                       OHSP  12 

C     USAGE                                                             OHSP  13 

C       CALL OHSP(EM,R,H,T,SV,ICALL)                                    OHSP  14 

C                                                                       OHSP  15 

C     DESCRIPTION OF ARGUMENTS                                          OHSP  16 

C       EM    - MAGNITUDE                                               OHSP  17 

C       R     - EPICENTRAL DISTANCE IN KILOMETERS                       OHSP  18 

C       H     - DAMPING FACTOR IN DECIMAL FRACTION                      OHSP  19 

C       T     - PERIOD IN SEC                                           OHSP  20 

C       SV    - SPECTRAL VELOCITY IN KINES                              OHSP  21 

C       ICALL - ICALL.EQ.0 FOR FIRST CALL                               OHSP  22 

C               ICALL.NE.0 FOR SUBSEQUENT CALLS WITH THE SAME EM,R,H    OHSP  23 

C                          AND DIFFERENT T                              OHSP  24 

C                                                                       OHSP  25 

C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED                     OHSP  26 

C       VELK                                                            OHSP  27 

C                                                                       OHSP  28 

Argument Type Parameter in calling program Return Parameter

EM R  Magnitude of earthquake  Unchanged

R R  Epicentral distance (unit : km)  Unchanged

H R  Damping factor  in decimal fraction  Unchanged

T R
 Period for calculating  velocity
 spectrum (unit: sec)  Unchanged

SV R  No need to input here  Spectral velocity  (unit: cm/sec)

ICALL I

 ICALL .EQ. 0 : First call
 ICALL .NE. 1 : Subsequent calls with
                           the same EM, R, H and
                           different T

 Unchanged

      SUBROUTINE OHSP(EM,R,H,T,SV,ICALL)                                OHSP  29 

C                                                                       OHSP  30 

      DIMENSION   D(3,3),W(3,3,8),U(3,3,8),UK(8)                        OHSP  31 

      EQUIVALENCE (UK(1),TB),(UK(2),TC),(UK(3),TD),(UK(4),SVA),         OHSP  32 

     *            (UK(5),SVB),(UK(6),SVC),(UK(7),SVD),(UK(8),SVE)       OHSP  33 

      SAVE        UK                                                    OHSP  34 

      DATA        D/5.,15.,60.,10.,45.,150.,25.,120.,350./,             OHSP  35 

     *            TA/.02/,TE/2./                                        OHSP  36 

      DATA        W/0.10,.12,.14,.20,.28,.28,.35,.45,.50,               OHSP  37 

     *              0.14,.25,.32,.40,.45,.50,.60,.70,.70,               OHSP  38 

     *              0.80,.65,.60,1.2,.90,.80,1.5,1.2,1.0,               OHSP  39 

     *              1.22,.76,.62,.52,.42,.46,.28,.30,.36,               OHSP  40 

     *               15., 9., 8.,12.,10.,11., 4.,10.,18.,               OHSP  41 

     *               21.,25.,27.,25.,27.,32.,30.,33.,38.,               OHSP  42 

     *               21.,25.,27.,25.,27.,32.,30.,33.,38.,               OHSP  43 

     *                9.,12.,15.,20.,24.,27.,28.,32.,37./               OHSP  44 

C                                                                       OHSP  45 

C     INITIALIZATION                                                    OHSP  46 

C                                                                       OHSP  47 

      IF(ICALL.NE.0) GO TO 210                                          OHSP  48 

      DO 130 K=1,8                                                      OHSP  49 

      DO 120 J=1,3                                                      OHSP  50 

      DO 110 I=1,3                                                      OHSP  51 

      U(I,J,K)=W(I,J,K)                                                 OHSP  52 

  110 CONTINUE                                                          OHSP  53 

  120 CONTINUE                                                          OHSP  54 

  130 CONTINUE                                                          OHSP  55 

C                                                                       OHSP  56 

C     EFFECT OF DAMPING FACTOR                                          OHSP  57 

C                                                                       OHSP  58 

      T0=10.**(0.31*EM-1.2)                                             OHSP  59 

      DO 160 K=5,8                                                      OHSP  60 

      DO 150 J=1,3                                                      OHSP  61 

      DO 140 I=1,3                                                      OHSP  62 

      U(I,J,K)=U(I,J,K)/SQRT(1.+17.*(H-0.05)*EXP(-2.5*U(I,J,K-4)/T0))   OHSP  63 

  140 CONTINUE                                                          OHSP  64 

  150 CONTINUE                                                          OHSP  65 

  160 CONTINUE                                                          OHSP  66 

C                                                                       OHSP  67 

C     INTERPOLATION                                                     OHSP  68 

C                                                                       OHSP  69 

      J=INT((EM+0.0001)/7.)+1                                           OHSP  70 

      EMR=EM-REAL(J+5)                                                  OHSP  71 

      R1=ALOG(D(2,J+1))*EMR+ALOG(D(2,J))*(1.-EMR)                       OHSP  72 

      RLOG=ALOG(R)                                                      OHSP  73 

      IF(RLOG-R1) 170,180,180                                           OHSP  74 

  170 I=1                                                               OHSP  75 

      R2=R1                                                             OHSP  76 

      R1=ALOG(D(1,J+1))*EMR+ALOG(D(1,J))*(1.-EMR)                       OHSP  77 

      GO TO 190                                                         OHSP  78 

  180 I=2                                                               OHSP  79 

      R2=ALOG(D(3,J+1))*EMR+ALOG(D(3,J))*(1.-EMR)                       OHSP  80 

  190 RR=(RLOG-R1)/(R2-R1)                                              OHSP  81 

      IF(RR.LT.0.) RR=0.                                                OHSP  82 

      VMULT=VELK(EM,R)/10.                                              OHSP  83 

      DO 200 K=1,8                                                      OHSP  84 

      U1=ALOG(U(I,J+1,K))*EMR+ALOG(U(I,J,K))*(1.-EMR)                   OHSP  85 

      U2=ALOG(U(I+1,J+1,K))*EMR+ALOG(U(I+1,J,K))*(1.-EMR)               OHSP  86 

      UK(K)=EXP(U2*RR+U1*(1.-RR))                                       OHSP  87 

      IF(K.LE.3) GO TO 200                                              OHSP  88 

      UK(K)=UK(K)*VMULT                                                 OHSP  89 

  200 CONTINUE                                                          OHSP  90 

C                                                                       OHSP  91 

C     SPECTRAL VELOCITY FOR GIVEN PERIOD                                OHSP  92 

C                                                                       OHSP  93 

  210 IF(T.GT.TA) GO TO 220                                             OHSP  94 

      SV=T/TA*SVA                                                       OHSP  95 

      RETURN                                                            OHSP  96 

  220 IF(T.GE.TB) GO TO 230                                             OHSP  97 

      SV=T*(SVA/TA+(T-TA)/(TB-TA)*(SVB/TB-SVA/TA))                      OHSP  98 

      RETURN                                                            OHSP  99 

  230 IF(T.GE.TC) GO TO 240                                             OHSP 100 

      SV=T*(SVB/TB+(T-TB)/(TC-TB)*(SVC/TC-SVB/TB))                      OHSP 101 

      RETURN                                                            OHSP 102 

  240 IF(T.GE.TD) GO TO 250                                             OHSP 103 

      SV=EXP(ALOG(SVC)+ALOG(T/TC)/ALOG(TD/TC)*ALOG(SVD/SVC))            OHSP 104 

      RETURN                                                            OHSP 105 

  250 SV=EXP(ALOG(SVD)+ALOG(T/TD)/ALOG(TE/TD)*ALOG(SVE/SVD))            OHSP 106 

      RETURN                                                            OHSP 107 

      END                                                               OHSP 108 

                                                                                                                        

 

 

【Example】 
 Calculate the Ohsaki’s spectrum with a damping factor of 5% for an earthquake of magnitude 7.3 with 
an epicentral distance of 25.0 km.  
 

C 

      DIMENSION T(51),SV(51) 

      DATA  N/51/,EM/7.3/,R/25.0/,H/0.05/ 

C 

      Q=ALOG(0.02) 

      P=(ALOG(2.0)-Q)/REAL(N-1) 

      DO 110 I=1,N 

      T(I)=EXP(REAL(I-1)*P+Q) 

      CALL OHSP(EM,R,H,T(I),SV(I),I-1) 

 110  CONTINUE 

      STOP 

C 

  601 FORMAT(F10.3,2X,F12.3 ) 

      END 

Output:  Calculate the spectral velocity for a period (stored in the array T) that divides logarithmically into 
50 equal parts between 0.02 and 2 sec. The computed Ohsaki’s spectrum is shown in the following 
figure. 

